

## **BlinkTask.nc** (konfiguráció)

---

```
configuration BlinkTask {
}
implementation {
    components Main, BlinkTaskM, SingleTimer, LedsC;

    Main.StdControl -> BlinkTaskM.StdControl;
    Main.StdControl -> SingleTimer;

    BlinkTaskM.Timer -> SingleTimer;
    BlinkTaskM.Leds -> LedsC;
}
```

## **BlinkTaskM.nc** (modul)

---

```
module BlinkTaskM {
    provides {
        interface StdControl;
    }
    uses {
        interface Timer;
        interface Leds;
    }
}
implementation {
    /**
     * the state of the red LED (on or off)
     */
    bool state;

    /**
     * Initialize the component.
     *
     * @return Always returns <code>SUCCESS</code>
     */
    command result_t StdControl.init() {
        state = FALSE;
        call Leds.init();
        return SUCCESS;
    }

    /**
     * Start things up. This sets the rate for the clock component.
     *
     * @return Always returns <code>SUCCESS</code>
     */
    command result_t StdControl.start() {
        return call Timer.start(TIMER_REPEAT, 1000);
    }
}
```

```

/**
 * Start things up. This just sets the rate for the clock component.
 *
 * @return Always returns <code>SUCCESS</code>
 **/
command result_t StdControl.stop() {
    return call Timer.stop();
}

/**
 * Module task: processing
 *
 * Toggles the red LED only.
 * The task is triggered by the <code>Timer.fired</code> event.
 *
 * @return None
 **/
task void processing()
{
    if (state)
        call Leds.redOn();
    else
        call Leds.redOff();
}

/**
 * Toggle the red LED in response to the <code>Timer.fired</code>
event
 * using the <code>processing</code> task.
 *
 * @return Always returns <code>SUCCESS</code>
 **/
event result_t Timer.fired()
{
    state = !state;
    post processing();
    return SUCCESS;
}
}

```

## SingleTimer.nc (konfiguráció)

```

configuration SingleTimer {
    provides interface Timer;
    provides interface StdControl;
}

implementation {
    components TimerC;

    Timer = TimerC.Timer[unique("Timer")];
    StdControl = TimerC;
}

```